Centrale 2007 (mais classique)
Déterminer les centres des cercles passant par l'origine et tangents à l'ellipse d'équation
$\frac{x^2}{5}+y^2=1$:
On commence par paramétrer une ellipse dans le cas général par

$M := [a \cdot \cos(t), b \cdot \sin(t)];$

$$[a\cos(t), b\sin(t)] \tag{1}$$

Le vecteur directeur est

$v := \mathit{diff}(M, t);$

$$[-a\sin(t), b\cos(t)] \tag{2}$$

Un vecteur normal à l'ellipse et unitaire est donc

$n := \left[\frac{v[2]}{\mathrm{sqrt}(v[1]^2+v[2]^2)}, -\frac{v[1]}{\mathrm{sqrt}(v[1]^2+v[2]^2)}\right];$

$$\left[\frac{b\cos(t)}{\sqrt{a^2\sin(t)^2+b^2\cos(t)^2}}, \frac{a\sin(t)}{\sqrt{a^2\sin(t)^2+b^2\cos(t)^2}}\right] \tag{3}$$

Le centre du cercle est donc, si R est le rayon

$\Omega := [M[1]+R \cdot n[1], M[2]+R \cdot n[2]];$

$$\left[a\cos(t)+\frac{R\,b\cos(t)}{\sqrt{a^2\sin(t)^2+b^2\cos(t)^2}}, b\sin(t)+\frac{R\,a\sin(t)}{\sqrt{a^2\sin(t)^2+b^2\cos(t)^2}}\right] \tag{4}$$

Le cercle passe par O si ce centre à la distance R de O

$R := \mathit{solve}(R^2=\mathrm{Omega}[1]^2+\mathrm{Omega}[2]^2, R);$

$$-\frac{1}{2}\,\frac{\cos(t)^2a^4\sin(t)^2+\cos(t)^4a^2b^2+\sin(t)^4b^2a^2+\sin(t)^2b^4\cos(t)^2}{a\sqrt{a^2\sin(t)^2+b^2\cos(t)^2}\;b\,\left(\cos(t)^2+\sin(t)^2\right)} \tag{5}$$

$R := \mathit{factor}(R);$

$$-\frac{1}{2}\,\frac{\left(b^2\sin(t)^2+\cos(t)^2a^2\right)\sqrt{a^2\sin(t)^2+b^2\cos(t)^2}}{a\,b\,\left(\cos(t)^2+\sin(t)^2\right)} \tag{6}$$

D'où les coordonnées du centre

$\mathit{simplify}(\Omega);$

$$\left[-\frac{1}{2}\,\frac{\left(\cos(t)^2a^2-2a^2+b^2-b^2\cos(t)^2\right)\cos(t)}{a},\; -\frac{1}{2}\,\frac{\left(-b^2\cos(t)^2+\cos(t)^2a^2-b^2\right)\sin(t)}{b}\right] \tag{7}$$

Dans l'exercice posé, on a

$a := \frac{1}{\sqrt{5}};\ b := 1;$

$$\frac{1}{5}\sqrt{5}$$

$$1 \tag{8}$$

Le rayon R dépend de t et vaut

$R;$

$$-\frac{1}{2}\,\frac{\left(\sin(t)^2+\frac{1}{5}\cos(t)^2\right)\sqrt{\frac{1}{5}\sin(t)^2+\cos(t)^2}\,\sqrt{5}}{\cos(t)^2+\sin(t)^2} \tag{9}$$

On définit une fonction ra qui donne le rayon en fonction de t et une fonction om pour le centre du cercle

$ra := unapply(R, t);$

$$t \to -\frac{1}{2}\,\frac{\left(\sin(t)^2+\frac{1}{5}\cos(t)^2\right)\sqrt{\frac{1}{5}\sin(t)^2+\cos(t)^2}\,\sqrt{5}}{\cos(t)^2+\sin(t)^2} \tag{10}$$

$om := unapply(\Omega, t);$

$$t \to \left[\frac{1}{5}\sqrt{5}\cos(t)-\frac{1}{2}\,\frac{\left(\sin(t)^2+\frac{1}{5}\cos(t)^2\right)\sqrt{5}\cos(t)}{\cos(t)^2+\sin(t)^2}, \sin(t)-\frac{1}{2}\,\frac{\left(\sin(t)^2+\frac{1}{5}\cos(t)^2\right)\sin(t)}{\cos(t)^2+\sin(t)^2}\right] \tag{11}$$

Reste à faire un dessin

```
with(plots) :
pc := u → plot([om(u)[1] + cos(t) * ra(u), om(u)[2] + sin(t) * ra(u), t = 0 ..2 * Pi], linestyle
    = 3) :
pc2 := u → plot([om(u)[1] + cos(t) * 0.01, om(u)[2] + sin(t) * 0.01, t = 0 ..2 * Pi], thickness
    = 3) :
p1 := plot([om(t)[1], om(t)[2], t = 0 ..2 * Pi], color = blue, thickness = 2) :
p2 := plot([cos(t) / sqrt(5), sin(t), t = 0 ..2 * Pi], color = black, scaling = constrained, thickness
    = 2) :
display(p1, p2, pc(Pi / 4), pc2(Pi / 4), pc(0), pc2(0), pc(Pi / 12), pc2(Pi / 12), pc(-Pi / 2), pc2(-Pi
    / 2));
```


1,0
0,5
-0,6
-0,4
-0,2
0
0,2
0,4
-0,5
-1,0